
Laborator 8. Arduino şi aplicaŃii WiFi - IoT 
 

IoT - Internet of Things se referă la conectarea diferitelor dispozitivelor electronice inteligente 
(inclusiv microcontrollere, senzori) la Internet. În această lucrare de laborator vom utiliza Arduino 
împreună cu un modul ESP8266 WiFi module. Cu acest dispozitiv, vom controla, dintr-o pagină web 
găzduită pe Arduino, starea unui LED de pe placă, şi vom vizualiza diferite informaŃii. 

ESP 8266 este un modul autonom WiFi care poate fi programat folosind mediul de dezvoltare 
Arduino. Programarea acestui dispozitiv poate fi realizată folosind un programator FTDI, sau o placă 
Arduino cu chipul ATMega328 detaşat. În acest laborator, vom configura esp8266 prin interfaŃa 
lui serială, folosind comenzi AT. 
 
Conectarea modulului WiFi 
 

Vom conecta modulul ESP la Arduino folosind interfaŃa UART Serial1. ConectaŃi pinul RX 
al ESP8266 la pinul 18 (TX1) Arduino, şi pinul TX al ESP8266 la pinul 19 (RX1). ConectaŃi pinul 
GND la un pin GND al Arduino, şi pinul VCC/3V3 la pinul Arduino de 3.3 V. ConectaŃi şi pinul EN 
al ESP8266 tot la 3.3 V. 
 
ConsultaŃi schema de mai jos pentru realizarea conexiunilor: 
 

 
 

Programul Arduino trebuie să genereze comenzi AT pentru resetarea modulului WiFi 
(“AT+RST”). Următorul pas este configurarea acestui modul ca punct de acces WiFi 
(“AT+CWMODE=2”). După acest pas, se citeşte adresa IP a modulului, 192.168.4.1, folosind 
comanda: “AT+CIFSR”, care va tipări şi adresa MAC.  



Pentru a obŃine informaŃia de conectare (SSID), vom rula comanda (“AT+CWSAP?”): se va 
returna numele reŃelei şi parola (implicit nu există parolă), şi apoi putem configura sistemul să accepte 
conexiuni multiple (“AT+CIPMUX=1”) şi vom porni serverul web pe portul 80 
(“AT+CIPSERVER=1,80”). 

Fiecare comandă AT trebuie terminată cu carriage return şi newline (“\r\n”). 
 
Mai multe informaŃii privind comenzile AT ale ESP8266 pot fi găsite aici: 

https://www.espressif.com/sites/default/files/documentation/4a-esp8266_at_instruction_set_en.pdf şi 
aici: https://github.com/espressif/esp8266_at/wiki. 
 

Comenzile sunt trimise folosind interfaŃa serială iniŃializată din programul Arduino 
(“Serial1”), folosind metoda “print”. Răspunsul modulului la comandă este citit şi salvat într-un 
string, şi apoi este afişat pe serial monitor, folosind conexiunea dintre Arduino şi PC, “Serial”. 
ConsultaŃi funcŃia “sendData()” din exemplul de mai jos. 

 
În bucla principală verificăm dacă datele sunt disponibile pe interfaŃa Serial1, şi verificăm 

dacă sunt date de pe reŃea (vor include substring-ul “+IPD”). Prima dată trebuie să citim 
identificatorul conexiunii, deoarece acesta este necesar când transmitem date folosind comanda: 
“AT+CIPSEND”.  

O pagină web este construită sub forma unui string şi trimisă la modulul ESP8266. Pagina 
include text de afişat, două butoane pentru preluarea comenzilor de la utilizator, şi apoi alt text de 
afişat, ce include date generate de Arduino. După ce se transmite comanda AT pentru transmiterea 
paginii web, se închide conexiunea folosind: “AT+CIPCLOSE”. 

Mecanismul pentru controlul led-ului pe Arduino se bazează pe folosirea a două butoane în 
pagina web, fiecare indicând un URL diferit. Primul buton va avea ataşat URL-ul “/l0” , iar al doilea 
“/l1”. Prin apăsarea acestor butoane, pagina web încearcă să redirecteze către aceste adrese, şi va 
genera o cerere către server. Pe Arduino, vom primi această cerere citind datele transmise de ESP8266 
(în funcŃia “sendData()”) şi vom verifica dacă conŃine substring-urile “/l0” sau “/l1” (ex: 
“response.indexOf("/l0") != -1“). 
 

Datele de la Arduino sunt afişate pe pagina web adăugând rezultatul funcŃiei “readSensor()” 
la string-ul paginii web, şi trimiterea lui folosind comenzi AT către modulul WiFi. În exemplul de mai 
jos, funcŃia va afişa de fapt rezultatul apelării funcŃiei millis().  
 
 
ESP 8266 cod exemplu: 
 

 
#define DEBUG true 
 
void setup() { 
  Serial.begin(115200); 
  Serial1.begin(115200); 
  pinMode(LED_BUILTIN, OUTPUT); 
  digitalWrite(LED_BUILTIN, LOW); 
  sendData("AT+RST\r\n", 2000, false); // resetare modul 
  sendData("AT+CWMODE=2\r\n", 1000, false); // configurare ca access point 
  sendData("AT+CIFSR\r\n", 1000, DEBUG); // citeste adresa IP 



  sendData("AT+CWSAP?\r\n", 2000, DEBUG); // citeste informatia SSID (nume retea) 
  sendData("AT+CIPMUX=1\r\n", 1000, false); // configurare conexiuni multiple 
  sendData("AT+CIPSERVER=1,80\r\n", 1000, false); // pornire server pe port 80 
} 
 
void loop() { 
  if (Serial1.available()) { 
    if (Serial1.find("+IPD,")) { 
      delay(500); 
      int connectionId = Serial1.read() - 48; // functia read() returneaza valori zecimale ASCII 
      // si caracterul ‘0’ are codul ASCII 48 
      String webpage = "<h1>Hello World!</h1><a href=\"/l0\"><button>ON</button></a>"; 
      String cipSend = "AT+CIPSEND="; 
      cipSend += connectionId; 
      cipSend += ",";  
      webpage += "<a href=\"/l1\"><button>OFF</button></a>"; 
       
      if (readSensor() > 0) { 
        webpage += "<h2>Millis:</h2>"; 
        webpage += readSensor(); 
 
      } 
 
      cipSend += webpage.length(); 
      cipSend += "\r\n"; 
      sendData(cipSend, 100, DEBUG); 
      sendData(webpage, 150, DEBUG); 
       
      String closeCommand = "AT+CIPCLOSE="; 
      closeCommand += connectionId; //se adauga identificatorul conexiunii 
      closeCommand += "\r\n"; 
      sendData(closeCommand, 300, DEBUG); 
    } 
  } 
} 
 
String sendData(String command, const int timeout, boolean debug) { 
  String response = ""; 
  Serial1.print(command); // trimite comanda la esp8266 
  long int time = millis(); 
  while ((time + timeout) > millis()) { 
    while (Serial1.available()) { 
      char c = Serial1.read(); // citeste caracter urmator 
      response += c; 
    } 
  } 
  if (response.indexOf("/l0") != -1) { 
    digitalWrite(LED_BUILTIN, HIGH); 
  } 
  if (response.indexOf("/l1") != -1) { 



    digitalWrite(LED_BUILTIN, LOW); 
  } 
  if (debug) { 
    Serial.print(response); 
  } 
  return response; 
} 
 
 
unsigned long readSensor() { 
  return millis(); 
} 
 
  
Rularea programului 
 

ÎncărcaŃi codul pe Arduino. AsiguraŃi-vă că modulul ESP8266 este alimentat la 3.3V, şi că 
pinul EN este conectat tot la 3.3V. Nu conectaŃi aceşti pini la 5V, pentru că veŃi distruge 
adaptorul! 
 

Un dispozitiv client convenabil pentru accesarea reŃelei este telefonul dvs. ScanaŃi după reŃele 
WiFi, şi găsiŃi numele reŃelei care se potriveşte cu cel afişat pe Serial Monitor.  
 

 
 
 

Nu uitaŃi să schimbaŃi baud rate-ul din serial monitor la 115200! După cum se observă, 
numele reŃelei din figura de mai sus este “AT-THINKER_3DC39D”, fără parolă. Acelaşi nume 
trebuie să fie vizibil pe dispozitivul dvs. mobil:  
 



             
 

Stânga: numele reŃelei WiFi a modulului ESP8266, afişat pe telefonul mobil. Dreapta: pagina web cu 
butoane, afişând valoarea millis(). 

 
Notă: fiecare adaptor are numele lui unic de reŃea. ConectaŃi-vă la reŃeaua adaptorului vostru propriu, 
nu la reŃelele colegilor ! 
 

După ce vă conectaŃi la reŃeaua WiFi, deschideŃi pe telefonul mobil un browser web, şi scrieŃi 
adresa serverului web: 192.168.4.1. Rezultatul ar trebui să fie cel din imaginea de mai sus, partea 
dreaptă. 
 
 
Lucru individual: 
 
1. RulaŃi exemplul. AsiguraŃi-vă că modulul WiFi este corect conectat, şi alimentat la 3.3V. 
 
2. ModificaŃi exemplul pentru a afişa date suplimentare în browser-ul web. ConectaŃi un senzor la 
Arduino, şi, folosind butoanele de pe pagina web, selectaŃi informaŃia de afişat (millis() sau datele de 
la senzor). AsiguraŃi-vă că pagina web informează clar utilizatorul ce fel de informaŃie este afişată. 
 
3. Folosind manualul modulului WiFi, schimbaŃi setările reŃelei WiFi: schimbaŃi numele reŃelei 
(SSID), adăugaŃi o parolă şi un mod de criptare. 
 


